THE MICRO WORKS

DIGISECTOR

DS-80

OWNER'S MANUAL

P.O. BOX 1110 DEL MAR, CA. 92014 714-756-2687

DS-80 DIGISECTOR BOARD

TABLE OF CONTENTS

Warranty and Repairs 1

Introduction 2

Unpacking and Installation 2

How It Works

- Sync Stripper & Video Amplifier 3
- Analog to Digital Converter 3
- Position Counters 4
- Control Circuitry 5
- Interface Logic 5
- Programming Notes 6

Test Routine 7

Adjustments and Test Points 8

Parts List 9

Schematics 11

Source Listings

- Command Processor 13
- Vector Graphics Subroutine 15
- ASCII Printer Subroutine 17
- Malibu Printer Subroutine 19

LIMITED WARRANTY

The Micro Works warrants its products to be free from defects in workmanship and materials for a period of ninety (90) days from the date of purchase. IT IS EXPRESSLY AGREED THAT THIS NINETY (90) DAY WARRANTY SHALL BE IN LIEU OF OTHER EXPRESS WARRANTIES, WARRANTIES OF FITNESS AND IN LIEU OF THE WARRANTY OF MERCHANTABILITY. No agent, representative or employee of the Company has authority to increase or alter the obligation of this warranty.

This warranty shall not apply to any Micro Works product which has been modified, repaired or altered in any way.

This warranty shall not apply to any Micro Works product damaged as a result of abuse, misuse, accident or neglect.

IN NO EVENT SHALL THE MICRO WORKS BE LIABLE FOR CONSEQUENTIAL DAMAGES.

IN ORDER TO MAKE A CLAIM AGAINST THIS WARRANTY THE DEFECTIVE BOARD MUST BE RETURNED BY PRIVATE CARRIER OR THE U.S. POSTAL SERVICE TO THE MICRO WORKS, P.O. BOX 1110, DEL MAR, CALIFORNIA, 92014. BOARDS MUST BE ACCOMPANIED BY RETURN SHIPPING CHARGES AND THE SALES RECEIPT SHOWING DATE OF PURCHASE. It is suggested that boards shipped through the United States mails be insured.

REPAIRS

At any time after the 90-day warranty period, The Micro Works will repair your PC board for a fee of $25.00, provided that it is not physically damaged, and not more than two chips need replacing. If this fee is not applicable, you will be notified before further repairs are made. If repairs are necessary, re-pack the board carefully and enclose a check to The Micro Works, P.O. Box 1110, Del Mar, CA, 92014.

INTRODUCTION

Thank you for purchasing a Micro Works Digisector. Every effort has been made in the development of the DS-80 to provide you with a long lasting, trouble free computer accessory. We suggest you read this manual thoroughly before installing the DS-80.

The Micro Works DS-80 Digisector is a video digitizer which accepts input from a video source such as a closed circuit television camera, converts the analog video signal to digital data, and transfers this data to the computer under software control. The DS-80 can resolve 256 X 256 picture elements (pixels) and provide up to 64 levels of grey scale. Conversion time for 6 bits is approximately 12 usec. The DS-80 accepts either standard (NTSC) or non-standard (industrial) input from the video source. Its output to the system monitor is the video plus an intensified cursor located at the point where the Digisector is currently looking.

Software provided in this manual transfers the output from the DS-80 to memory and prints a picture reflecting the digitized brightness of each point stored in memory. This software is written specifically to drive a Malibu Design Group Model 160 printer and will require modification for other printers. Print routines unique to the Malibu printer are noted in the listings.

The Micro Works is certain you will find the DS-80 a true enhancement to your computer system. We look forward to hearing any suggestions or comments from our customers and are interested in the various applications to which the Digisector will be put.

UNPACKING AND INSTALLATION

Carefully remove the DS-80 from the box and unwrap the packing material. Take time to inspect the PC board for any damage which may have been incurred in shipping. If there is any damage, save all packing materials and notify the carrier immediately.

Your DS-80 contains MOS integrated circuitry which may be easily damaged by static electrical sources. Avoid overhandling and do not allow anything to come into contact with the conductors on the board. Never lift the board out of, or plug it into, a computer which is turned on. We urge you to make sure that your S-100 computer system has been completely tested before you install the DS-80.

Bring the camera output and monitor input cables through a hole in the computer cabinet and attach to the Molex connector on the top of the DS-80 board. The camera input to the Digisector is labelled VI, and the video monitor output is labelled VO on the board, for connection to the camera and monitor respectively.

HOW IT WORKS

Sync Stripper and Video Amplifier

Composite video appearing at the input is amplified to 2 volts peak-to-peak by U15. DC restoration clamps the sync tips to ground through Ql. This clamp pulse, appearing at the collector of Ql, is sliced by Q2 and applied to sync stripper logic U20 and U21. This provides separated horizontal and vertical sync pulses, VSYNC, HBLANK and HBLANK. These pulses control the horizontal and vertical position counters and will be discussed later.

Analog to Digital Converter

The amplified video is connected to a sample and hold circuit consisting of Q3, C5 and buffer amplifier U14. When the sample line is pulsed high, Q3 conducts, charging C5 up to the value of the video signal at that instant. Since amplifier U14 has FET inputs, leakage current is low and the cap remembers this voltage level during subsequent A-D conversion.

Ull, U12 and U13 comprise a high speed successive approximation A-D converter. Most of the smarts in this converter are contained in the

successive approximation register (SAR) Ull, which controls the conversion sequence. Conversion is accomplished as follows: R18 sets a DC level of about .6 volt at pin 3 of comparator U13. The signal level to be digitized is present at the output of buffer amplifier U14. The SAR controls a current output D-A converter, U12; all bits are initially off.

The SAR first turns on the most significant bit of this D-A converter, causing 1 ma. of current flow through R20 and R17. If this current is sufficient to drop the voltage at pin 2 of the comparator to below the threshold voltage at pin 3, the comparator output goes high, telling the SAR that that's too much current. In that case the SAR turns the MSB back off and turns on the next most significant bit. This process continues for all 6 bits; if the comparator output stays low after a given bit is turned on, the SAR leaves that bit on. If it goes high, the SAR turns it off again before proceeding to the next most significant bit.

In this way, the SAR sneaks up on the correct value of current to supply to the summing junction at pin 2 of U13 so that when the conversion is done, pin 2 is within a few millivolts of the threshold voltage at pin 3, and the binary weighted bits that supplied the current necessary to "null out" the input voltage are present at the outputs of the SAR. This binary number is the digitized value of the brightness level at the sample point. The brightness adjustment, R18, is set to make the "0" brightness level equal to the blanking, or black level of the incoming video. The contrast adjustment, R17, sets the gain of the A-D converter by varying the amount of current per volt supplied to U13 pin 2. Since the conversion is now complete, the data is present at the inputs of buffer U10, ready for reading by the computer.

Position Counters

U8, U9 and U21 comprise a self latching counter with a maximum count of 255; that is, when started, it counts up to 255 and then quits until it is reset again. This counter is reset by the occurrence of the vertical sync pulse (60 Hz.) and incremented by the horizontal sync pulses (15,750 Hz.)

and as such, contains the vertical position (in scan lines) of the scanning spot at all times. This is our "Y" coordinate. U6, U7 and U26 make up a similar counter, but this one is reset by the horizontal sync and incremented by the high speed dot clock composed of U20 and gates U22 and U25. The frequency of this dot clock determines the width of the picture scanned and is adjustable by R16. This counter contains our "X" coordinate.

Control Circuitry

We now have two counters which keep track of the X and Y coordinates of the scanning spot in the incoming video at all times, and a means of digitizing the brightness of any individual spot. Getting them together is easy. The outputs of the X and Y position counters are connected to one set of inputs of a 16 bit digital comparator, U2-U5. The other inputs to this comparator are connected to the two 8 bit latches, UO and Ul. When the scanning spot coordinates (counter outputs) are equal to the desired X-Y coordinates (latch outputs), a 200 ns. wide EQU pulse is produced by the comparator. This pulse signifies that the dot is in exactly the spot we wish to digitize, so we use it to provide the sample pulse to the sample and hold circuit, capturing the brightness of the spot we want. This pulse also triggers the SAR to begin its A-D conversion, as described before, and is summed with the video output to provide a cursor which shows where the Digisector is looking. U19 and associated gates complete the handshaking necessary for DS-80 operation.

Interface Logic

The DS-80 occupies 2 contiguous DIP switch selectable I/O ports. U17 and U18 compare the setting of the port select DIP switch to the lower 8 bits of the address lines. During I/O operations, this produces I/O read and write signals corresponding to the selected port address. Allocation is as follows:

	OUTPUT	INPUT
Even Port	X ADDR	HALT / READ DATA
Odd Port	Y ADDR	READ STATUS / READ DATA

An input at the even port halts the CPU until the acquisition and conversion of data is complete. An input at the odd port will return the MSB = 1 if conversion is incomplete, but will not halt the processor.

Programming notes

To digitize a point's brightness, simply output the X ADDR (0 - 255) desired to the even port; then output the Y ADDR (0 - 255) to the odd. A write to the Y address port automatically commands the DS-80 to go digitize a point. Two methods are available for synchronizing the CPU to the DS-80: status and halt mode. Examples are given of both.

```
WAIT:   IN     PORT+1
        JM     WAIT
               ;Data is now in A

HALT:   IN     PORT
               (Processor halts if data is not available.)
               ;Data is now in A
```

The Digisector's speed is generally software limited when sampling blocks of video data (such as the portrait program provided). When randomly addressing the video, remember that a given point on the raster occurs only once every 16.66 ms. and plan digitizing algorithms accordingly. For example, if the following points are desired:

(1,1), (11,11), (21,21), (31,31), (41,41), (51,51)

they should be acquired in ascending order, as written. Time for acquisition would be (8 ms. average latency) + 5 ms. = 13 ms. If they were to be acquired in the reverse order, acquisition would take (8 ms. average latency) + 5 X 16 ms. = 88 ms. Such is the nature of raster scan devices. Remember that the upper left hand corner of the screen is (1,1) and the lower right is (FF,FF).

Due to the latching nature of the X and Y position counters, the address $0,0,_{16}$ is illegal and will produce undefined results, since this

state occurs for a much longer time than one picture element (200 ns.).

We have found that proper lighting is the most important factor in obtaining quality images using the DS-80.

The following short program may be used as a test and familiarization aid in operating the DS-80. This program continuously samples the center point (80_{16},80_{16}) on the TV screen and sends the brightness value to the user's console device in hexadecimal form. The cursor may be observed and experiments with lighting and operation may be performed while this routine is running.

```
                ;*****************************************************************
                ;*                                                               *
                ;*   THIS ROUTINE IS TO TEST YOUR DIGISECTOR.                    *
                ; THIS ROUTINE WILL SET THE X AND Y ADDRESS TO 80 ON            *
                ; THE DS-80 START THE CONVERSION AND OUTPUT IT TO YOUR          *
                ; PROGRAMMED OUTOUT LIGHTS. FOR THE PEOPLE WHO DON'T HAVE+
                ; PROGRAMMED OUTPUT LIGHTS YOU SHOULD EITHER SKIP THIS         *
                ; PROCESS OR WRITE A ROUTINE TO CONVERT THE DATA TO HEX
                ; AND OUTPUT THIS TO YOUR CONSOLE.                             *
                ;                                                               *
                ;       NOT COPYRIGHTED 1979 BY THE MICRO WORKS                 *
                ;                                          BY MIKE LESHER       *
                ;*****************************************************************
0100                    ORG     100H
          0050  XPORT:  EQU     50H         ;X OUTPUT PORT
          0051  YPORT:  EQU     51H         ;Y OUTPUT PORT AND START CONVERSION
          0050  INPORT: EQU     50H         ;TO GET DATA
          00FF  LIGHTS: EQU     0FFH        ;PROGRAMMED OUTPUT LIGHTS

0100 3E 80      START:  MVI     A,80H
0102 D3 50              OUT     XPORT
0104 D3 51              OUT     YPORT       ;SET CURSOR TO MIDDLE OF SCREEN.
0106 DB 50              IN      INPORT      ;HALT UNTIL DATA READY AND PUT IN A.
0108 D3 FF              OUT     LIGHTS      ;PUT ON LIGHTS
010A C3 00 01           JMP     START       ;DO IT AGAIN
                ;
                ; SIMPLE WASN'T IT.
                ;
                        END

      0 ERROR(S) DETECTED
        LAST ADDRESS 010C


SYMBOL TABLE:

INPORT 0050    LIGHTS 00FF    START  0100    XPORT  0050    YPORT  0051
```

ADJUSTMENTS AND TEST POINTS

Scope Settings: (See Fig. 1) 20 usec./Div. 1v./Div. DC coupled

Width:

Connect probe #1 to pad marked VIDEO. This is amplified video, and should be a composite video signal of 2 v. peak-to-peak, with sync tips at ground.
Trigger scope on horizontal sync pulse.
Connect probe #2 to U20, pin 5. This is horizontal dot clock.
Adjust R16 until the horizontal dot clock pulses cover active video area on horizontal line.

If no scope is available, adjust for correct aspect ratio of output.

Brightness:

Connect probe #1 to pad marked VIDEO.
Connect probe #2 to pad marked REF. This is brightness reference level.
Adjust pot R18 until brightness reference level is equal to minimum video level present during active video portion of horizontal line.

If no scope is available, adjust for preferred brightness.

NOTE: Increasing the reference level will darken the picture, while reducing the level will lighten it. Approximately .6 volt is a good starting point.

Contrast:

Adjust pot R17 for desired contrast.

Sample:

The pad marked SAMPLE will show the sample pulse during operation.

A/D Test:

The pad marked A/D Test is the output of the comparator to the SAR; this can be used to test the A/D circuitry if necessary. (See pg. 4.)

ONE HORIZONTAL LINE SHOWN



Figure 1.

DS-80 PARTS LIST

Integrated Circuits

U0 and Ul	74LS273
U2 through U5	74LS85
116 through U9	74LS161
U10	74LS367
Ull	MC14559
U12	1408-L6
U13	LM311N
U14	CA3140
U15	LM318
U16	74LS125
U17 and U18	74LS85
U19	74LS138
U20	74LS123
U21 and U26	74LS74
U22	74LS00
U23	74LS04
U24	74LS132
U25	74LS02
U27	7805
U28	7912
U29	7812

Resistors (Ohms)

Rl	220
R2, R6 and R9	75
R3, R14, R15, R22 and R23	2.2K
R4	22K
R5 and Rll	10K
R7	6.8K
R8	5.6K
R10	2.2 Meg
R12, R21, R27 through R35	1K
R13	470
R16	2K Trim (Width)
R17	500 Trim (Contrast)
R18	2K Trim (Brightness)
R19	4.7K
R20	100
R24, R25 and R26	1.5K

Capacitors

Cl, C6, C10, Cll, C13 and C15	.1 mf
C2	.005 mf
C3	30 pf.
C4	160 pf.
C5	220 pf.
C7	20 pf.
C8	.001 mf.

Capacitors (cont.)

C9	.01 mf.
C12	100 mf./16V.
C14 and C16	22 mf./35V.

Diodes

CR1 through CR6	1N914

Transistors

Ql	2N3906
Q2 and Q4	2N3904
Q3	SD210


VIDEO OUT
+12V
R3 2.2K
CR1
Q1 2N 3906
MARK
R1 220Ω
R2 75Ω
Q3 SD 210
SAMPLE
C5 220PF
+5V
CA 3140 U14
S & H
R17 500Ω
CONTRAST
R20 100Ω
BRIGHT LEVEL
R18 2K
R19 4.7K
REF
C6 .1uF
LM311 U13
R21 1K
A/D TEST
Q4 2N 3904
R27 1K
R26 1.5K
C10 .1uF
U12 1408-L6
R22 2.2K
C7 70PF
-12V
U10 36T
DI5 92
DI4 91
DI3 42
DI2 41
DI1 94
DI0 95
1,15 G DATA
VIDEO IN
C1 .1uF
CR2
CR3
R10 2.2M
R9 75Ω
LM318 U15
R15 2.2K
VIDEO
C4 100 PF
R14 2.2K
R4 22K
U11 14559
U24 132
R24 1.5K
C8 .001uF
A/D & DATA
R11 10K
Q2 2N 3904
R5 10K
R6 75Ω
CR5
R7 6.8K
C2 .005
U20 123
U21(b) 74
VSYNC
H BLANK
H BLANK
START
EOC
R25 1.5K
CR6
U23 04
C9 .01 uF
R23 2.2K
U26(c) 74
R12 1K
CR4
R13 470Ω
U25 02
MARK
SAMPLE
EQU


R8
5.6K
R16 WIDTH
2K
+5V
H BLANK
HORIZONTAL
VERTICAL
C3
.30PF
U8
U9
U6
U7
U5
U4
U3
U2
U1
U0
U26(d)
U21(a)
VSYNC
EQU
90 DO7
40 DO6
39 DO5
38 DO4
89 DO3
88 DO2
35 DO1
36 DO0
DIP SWITCH
R31
R32
R33
R34
R28
R29
R30
R35
1K
U18
U17
U16
EOC
PRDY 72
XSCL
START
U19
138
YSCL
HALT
STAT
G DATA
ADDRESS
A7 A6 A5 A4
83 82 29 90
A3 A2 A1
31 81 80
SINP SOUT A0
46 45 79
PWR
77
PDB IN
78
DI6 DI7
93 94
THE MICRO WORKS
DIGISECTOR
DS-80
©1979

```
                ;*****************************************************
                ;*                                                   *
                ;*  THIS ROUTINE LINKS THE OTHER PORTRAIT ROUTINES BY *
                ;* ALLOWING THE USER TO TYPE A CHARACTER AND INVOKING *
                ;* THE APPROPRATE ROUTINE.                           *
                ;*                                                   *
                ;*      COPYRIGHT 1979 BY THE MICRO WORKS            *
                ;*                                   BY MIKE LESHER  *
                ;*****************************************************

0100                    ORG     100H
        0200    NSCAN:  EQU     200H        ;NSCAN ROUTINE ADDRESS
        0300    PRINT:  EQU     300H        ;PRINTER DRIVER
        000D    CR:     EQU     0DH
        000A    LF:     EQU     0AH
        0000    NULL:   EQU     00H
        0000    MONITOR: EQU    0000H       ;SYSTEM MONITOR

0100 C3 31 01   INIT:   JMP     MESAGE
0103 21 7D 01   WARM:   LXI     H,PROMPT    ;POINT TO PROMPT
0106 CD 3D 01           CALL    PMESAGE     ;PRINT IT
0109 CD 57 01           CALL    CIN         ;GET A CHAR
010C E6 7F              ANI     7FH         ;MASK OFF THE DATA
010E F5                 PUSH    PSW         ;SAVE CHAR
010F 4F                 MOV     C,A
0110 CD 54 01           CALL    COUT        ;ECHO THE CHAR
0113 CD 49 01           CALL    CRLF        ;DO A CR, LF
0116 F1                 POP     PSW         ;RESTORE CHAR
0117 FE 54              CPI     'T'         ;IS IT A 'T'?
0119 CA 00 02           JZ      NSCAN       ;YES, JMP NSCAN
011C FE 50              CPI     'P'         ;IS IT A 'P'?
011E CA 00 03           JZ      PRINT       ;YES, JMP TO PRINT ROUTINE
0121 FE 4D              CPI     'M'         ;IS IT AN 'M'?
0123 CA 00 00           JZ      MONITOR     ;YES, JUMP TO MONITOR
0126 0E 3F              MVI     C,'?'       ;ERROR
0128 CD 54 01           CALL    COUT        ;OUTPUT IT
012B CD 49 01           CALL    CRLF
012E C3 03 01           JMP     WARM        ;NOT FOUND, PROMPT AGAIN.
0131 31 C5 01   MESAGE: LXI     SP,STACK    ;SET UP STACK
0134 21 5A 01           LXI     H,NERF      ;WHO MADE THIS
0137 CD 3D 01           CALL    PMESAGE     ;PRINT IT
013A C3 03 01           JMP     WARM        ;SEND PROMPT
013D 7E         PMESAGE: MOV    A,M         ;GET CHAR
013E FE 00              CPI     NULL        ;IS IT A 00
0140 C8                 RZ                  ;YES, RETURN
0141 4F                 MOV     C,A         ;MOVE IT TO C
0142 CD 54 01           CALL    COUT        ;PRINT IT
0145 23                 INX     H           ;INCREMENT POINTER
0146 C3 3D 01           JMP     PMESAGE     ;SEND MORE
0149 0E 0D      CRLF:   MVI     C,CR        ;OUTPUT A CARRAGE RETURN
014B CD 54 01           CALL    COUT
014E 0E 0A              MVI     C,LF        ;OUTPUT A LINE FEED
```

```
0150 CD 54 01            CALL    COUT
0153 C9                  RET
                         *
                         *COUT IS A USER ROUTINE TO PRINT THE CHAR IN C REG. ON THE
                         *MASTER CONSOLE.  ONLY REGS. H, L, AND C MUST BE SAVED.
                         *CIN IS TO GET A CHAR FROM THE USERS KEYBOARD AND PUT IT IN
                         *THE A REG.  NO REGS. NEED TO BE SAVED.
                         *
0154 C3 09 F0  COUT:     JMP     0F009H
0157 C3 03 F0  CIN:      JMP     0F003H
015A 54 48 45  NERF:     DB      'THE MICRO WORKS',CR,LF
015D 20 4D 49
0160 43 52 4F
0163 20 57 4F
0166 52 4B 53
0169 0D 0A
016B 50 4F 52            DB      'PORTRAIT SYSTEM',CR,LF,NULL
016E 54 52 41
0171 49 54 20
0174 53 59 53
0177 54 45 4D
017A 0D 0A 00
017D 54 41 4B  PROMPT:   DB      'TAKE(T), PRINT(P) OR MONITOR(M)',NULL
0180 45 28 54
0183 29 2C 20
0186 50 52 49
0189 4E 54 28
018C 50 29 20
018F 4F 52 20
0192 4D 4F 4E
0195 49 54 4F
0198 52 28 4D
019B 29 00
019D                     DS      40
      01C5  STACK:       EQU     $
                         END     100H

    0 ERROR(S) DETECTED
      LAST ADDRESS 01C4
```

COUT
PUSH B
PUSH H
MOV E,C
MVI C,
CALL BDOS
POP H
POP B
RET

CIN
MVI C,
CALL BDOS
RET

SYMBOL TABLE:

```
CIN     0157    COUT    0154    CR      000D    CRLF    0149    INIT    0100
LF      000A    MESAGE  0131    MONITO  0000    NERF    015A    NSCAN   0200
NULL    0000    PMESAG  013D    PRINT   0300    PROMPT  017D    STACK   01C5
WARM    0103
```

```
                ********************************************************
                *                                                      *
                *  THIS ROUTINE CAUSES THE DS-80 TO DIGITIZE A 128X128 *
                * PICTURE AND SAVE IT IN AN 8K BLOCK OF MEMORY PACKED  *
                * 2 PIXELS PER BYTE. THE DATA IS SCANNED IN ROW MAJOR  *
                * ORDER; I.E. THE FIRST 64 BYTES CONTAIN THE UPPERMOST *
                * HORIZONTAL SET OF PIXELS, TO BE COMPATIBLE WITH THE  *
                * VECTOR GRAPHICS CO. HIGH RESOLUTION GRAPHICS DISPLAY *
                * BOARD.  IF THIS BOARD IS USED AS THE DISPLAY BUFFER  *
                * MEMORY IT WILL DISPLAY THE PICTURE AS IT IS SCANNED. *
                *                                                      *
                *       COPYRIGHT 1979 BY THE MICRO WORKS              *
                *                                    BY MIKE LESHER    *
                ********************************************************

0200                            ORG     200H
          8000  DISP:   EQU     8000H           ;DISPLAY/BUFFER
          0103  WARM:   EQU     0103H

0200 01 00 01   NSCAN:  LXI     B,0100H         ;START OF SCAN ON DIGISECTOR
0203 21 00 80           LXI     H,DISP          ;STORE START ADDRESS OF DISPLAY
0206 22 76 02           SHLD    LOC
0209 78         MORE:   MOV     A,B
020A D3 50              OUT     50H
020C CD 3B 02           CALL    HSCAN           ;GOTO HIGH NIBBLE SCAN
020F 04                 INR     B               ;X=X+2
0210 04                 INR     B
0211 78                 MOV     A,B
0212 D3 50              OUT     50H
0214 CD 58 02           CALL    LSCAN           ;LOW NIBBLE SCAN
0217 04                 INR     B               ;X=X+2
0218 04                 INR     B
0219 2A 76 02           LHLD    LOC             ;INCREMENT TOP OF SCREEN POINTER
021C 23                 INX     H
021D 22 76 02           SHLD    LOC
0220 78                 MOV     A,B
0221 FE 01              CPI     01H
0223 C2 09 02           JNZ     MORE
0226 3E 80              MVI     A,80H           ;PUT CURSOR AT MIDDLE OF SCREEN.
0228 D3 50              OUT     50H             ;WHEN NOT SCANNING.
022A D3 51              OUT     51H
022C 21 00 9E           LXI     H,DISP+1E00H ;POINT TO FIRST CHAR. IN 8TH
022F 36 FF      CLEAN:  MVI     M,0FFH          ;TO LAST LINE AND WHITE OUT
0231 23                 INX     H               ;UNTIL YOU GET TO THE BOTTOM.
0232 3E A0              MVI     A,DISP/256+20H ;A=END OF SCREEN
0234 BC                 CMP     H               ;A=H?
0235 C2 2F 02           JNZ     CLEAN           ;NO, CLEAN SOME MORE
0238 C3 03 01           JMP     WARM            ;GO BACK TO CENTRAL ROUTINE
                *
                *THIS ROUTINE MAKES A PASS ON THE DS-80 AND PUTS IT IN THE
                * HIGH NIBBLE.
```

```
023B D5          HSCAN:  PUSH    D
023C 3E 10               MVI     A,8*2       ;START 8 SCAN LINES DOWN ON DS-80
023E 81                  ADD     C           ;ADD TO C AND PUT BACK IN C
023F 4F                  MOV     C,A
0240 11 40 00            LXI     D,40H       ;NUMBER OF BYTES/LINE
0243 2A 76 02            LHLD    LOC         ;GET TOP OF SCREEN ADDRESS
0246 79          LOOP:   MOV     A,C         ;GET Y
0247 D3 51               OUT     51H         ;OUT TO Y PORT, START CONVERSION
0249 DB 50               IN      50H         ;HALT UNTIL EOC THEN GET DATA
024B 17                  RAL                 ;ROTATE LEFT TWICE
024C 17                  RAL
024D E6 F0               ANI     0F0H        ;MASK OFF NIBBLE
024F 77                  MOV     M,A         ;STORE IT IN MEMORY
0250 19                  DAD     D           ;HL=NEXT LINE DOWN
0251 0C                  INR     C           ;Y=Y+2
0252 0C                  INR     C
0253 C2 46 02            JNZ     LOOP        ;IF NOT TOP OF LINE GET MORE DATA
0256 D1                  POP     D
0257 C9                  RET                 ;RETURN TO MAIN ROUTINE
                 *
                 *THIS ROUTINE MAKES THE SCAN AFTER HSCAN, TAKING THE BYTE
                 *HSCAN USED AND PUTTING NEW DATA INTO THE LOWER NIBBLE.
                 *
0258 D5          LSCAN:  PUSH    D
0259 3E 10               MVI     A,8*2       ;START 8 LINES DOWN ON DIGISECTOR
025B 81                  ADD     C           ;ADD AND PUT BACK IN C
025C 4F                  MOV     C,A
025D 11 40 00            LXI     D,40H       ;NUMBER OF BYTES/LINE
0260 2A 76 02            LHLD    LOC         ;GET TOP OF SCREEN ADDRESS
0263 79          LLOOP:  MOV     A,C         ;GET Y
0264 D3 51               OUT     51H         ;OUT TO Y PORT, START CONVERSION
0266 DB 50               IN      50H         ;HALT UNTIL EOC THEN GET DATA
0268 1F                  RAR                 ;ROTATE RIGHT TWICE
0269 1F                  RAR
026A E6 0F               ANI     0FH         MASK OFF NIBBLE
026C B6                  ORA     M           ;OR MEMORY WITH A AND PUT IN A
026D 77                  MOV     M,A         ;MOVE BACK TO MEMORY
026E 19                  DAD     D           ;HL=NEXT LINE
026F 0C                  INR     C           ;Y=Y+2
0270 0C                  INR     C
0271 C2 63 02            JNZ     LLOOP       ;IF NOT BEGIN OF LINE GET MORE DATA
0274 D1                  POP     D
0275 C9                  RET                 ;RETURN TO MAIN ROUTINE
0276             LOC:    DS      2           ;TEMP STORAGE FOR HL
                         END

      0 ERROR(S) DETECTED
        LAST ADDRESS 0277


SYMBOL TABLE:

CLEAN   022F    DISP    8000    HSCAN   023B    LLOOP   0263    LOC     0276
LOOP    0246    LSCAN   0258    MORE    0209    NSCAN   0200    WARM    0103
```

```
                ********************************************************
                *                                                      *
                *  THIS ROUTINE WILL UNPACK THE DATA AFTER NSCAN HAS   *
                * DONE ITS WORK, USE A LOOKUP TABLE TO FIND THE        *
                * CORRESPONDING ASCII CHARACTER THEN CALL A USER       *
                * PRINT ROUTINE.                                       *
                *                                                      *
                *       COPYRIGHT 1979 BY THE MICRO WORKS              *
                *                                        BY MIKE LESHER *
                ********************************************************

0300                            ORG     300H

                8000    SCREEN: EQU     8000H       ;SCREEN/BUFFER
                0103    WARM:   EQU     0103H
                000D    CR:     EQU     0DH
                000A    LF:     EQU     0AH

0300 21 00 80   NPRINT: LXI     H,SCREEN    ;GET SCREEN ADDRESS
0303 06 40              MVI     B,40H       ;NUMBER OF BYTES/LINE
0305 7E         MAIN:   MOV     A,M         ;GET 2 PIXEL
0306 2F                 CMA
0307 E6 F0              ANI     0F0H        ;MASK OFF ONE NIBBLE
0309 1F                 RAR                 ;GET IT TO LOWER NIBBLE
030A 1F                 RAR
030B 1F                 RAR
030C 1F                 RAR
030D CD 33 03           CALL    LOOK        ;LOOKUP AND PRINT
0310 7E                 MOV     A,M         ;FOR 64 CHAR/LINE INSERT 'JMP SMALL'
0311 2F                 CMA
0312 E6 0F              ANI     0FH         ;GET LOWER PIXEL
0314 CD 33 03           CALL    LOOK        ;LOOKUP AND PRINT IT
0317 23         SMALL:  INX     H           ;POINTER=POINTER+1
0318 3E 9E              MVI     A,SCREEN/256+1EH ;TO END OF SCREEN?
031A BC                 CMP     H
031B CA 03 01           JZ      WARM        ;YES, GOTO MAIN ROUTINE
031E 05                 DCR     B
031F C2 05 03           JNZ     MAIN        ;IF NOT TO END OF LINE, PRINT MORE
0322 0E 0D      CRLF:   MVI     C,CR        ;PRINT CARRAGE RETURN
0324 CD 41 03           CALL    COUT
0327 0E 0A              MVI     C,LF        ;PRINT LINEFEED
0329 CD 41 03           CALL    COUT
032C 11 00 00           LXI     D,00H       ;USE 80H FOR 64 CHAR/LINE
032F 19                 DAD     D           ;THIS IS FOR ASPECT RATIO
0330 C3 03 03           JMP     MAIN-2
0333 E5         LOOK:   PUSH    H           ;SAVE HL
0334 21 44 03           LXI     H,TABLE     ;GET TABLE ADDRESS
0337 5F                 MOV     E,A         ;OFFSET=E
0338 16 00              MVI     D,00        ;D=0
033A 19                 DAD     D           ;HL=HL+DE
033B 4E                 MOV     C,M         ;PUT CHAR IN C
033C CD 41 03           CALL    COUT        ;OUTPUT IT
```

```
033F E1                 POP     H           ;RESTORE HL
0340 C9                 RET
0341 C3 0F 7E  COUT:    JMP     7E0FH       ;JUMP TO YOUR PRINT ROUTINE
0344 20 2E 2C  TABLE:   DB      ' .,:;-=+/?*%&$@#'     ;LEAST TO DARKEST PRINT
0347 3A 3B 2D
034A 3D 2B 2F
034D 3F 2A 25
0350 26 24 40
0353 23
                        END

      0 ERROR(S) DETECTED
        LAST ADDRESS 0353

SYMBOL TABLE:

COUT    0341    CR      000D    CRLF    0322    LF      000A    LOOK    0333
MAIN    0305    NPRINT  0300    SCREEN  8000    SMALL   0317    TABLE   0344
WARM    0103
```

```
                ********************************************************
                *                                                      *
                *   THIS ROUTINE IS FOR DRIVING THE MALIBU DESIGN GROUP *
                * PRINTER. IT WILL FIRE THE PRINT WIRES TO CREATE A GRAY *
                * SCALE PLUS MOVE THE HEAD BI-DIRECTIONALY DURING      *
                * PRINTING.                                            *
                *                                                      *
                *       COPYRIGHT 1979 BY THE MICRO WORKS              *
                *                                    BY MIKE LESHER    *
                ********************************************************


0300                            ORG     300H
          8000  DISP:   EQU     8000H       ;DISPLAY/BUFFER ADDRESS
          0002  LEFT:   EQU     2H          ;PRINT HEAD LEFT
          0001  RIGHT:  EQU     1H          ;   "   "    RIGHT
          0003  STOP:   EQU     3H          ;STOP PRINT HEAD
          0007  TOF:    EQU     7H          ;TOP OF FORM
          000A  FIRE:   EQU     0AH         ;FIRE PRINT HEAD
          0006  UP:     EQU     6H          ;MOVE PAPER UP
          0103  WARM:   EQU     0103H       ;BOOT ADDRESS
          0009  SENSE:  EQU     9
          0008  DATA:   EQU     8
          0009  CONTL:  EQU     9

0300 3E 40      MPRINT: MVI     A,40H       ;NUMBER OF LINES LEFT TO PRINT
0302 32 7B 04           STA     LINES
0305 21 00 80           LXI     H,DISP      ;HL=DISPLAY/BUFFER ADDRESS
0308 3E 02      HOME:   MVI     A,LEFT      ;START HEAD MOVING LEFT
030A CD A4 03           CALL    COMND1
030D DB 09      HOME1:  IN      SENSE       ;WAIT FOR COWS TO COME HOME
030F E6 04              ANI     4H
0311 C2 0D 03           JNZ     HOME1
0314 3E 03              MVI     A,STOP      ;LOCK BARN DOOR
0316 CD A4 03           CALL    COMND1
0319 3A 7B 04           LDA     LINES       ;LINES = LINES - 1
031C 3D                 DCR     A
031D 32 7B 04           STA     LINES
0320 CA DE 03           JZ      CLOSE       ;IF LINES=0 THEN DONE
0323 3E 01              MVI     A,RIGHT     ;START HEAD RIGHT
0325 CD A4 03           CALL    COMND1
0328 16 00              MVI     D,00        ;DIRECTION FOR PRINT ROUTINE=FWRD
032A CD BA 03           CALL    SWAIT       ;WAIT 3 SECTOR HOLES ON PRINTER
032D CD BA 03           CALL    SWAIT       ;TO MOVE PICTURE TO MIDDLE
0330 CD BA 03           CALL    SWAIT
0333 3E 80              MVI     A,80H       ;NUMBER OF CHAR. TO PRINT/PASS
0335 32 7A 04           STA     COUNT
0338 E5                 PUSH    H
0339 D5                 PUSH    D
033A 11 C0 1F           LXI     D,40H*128-40H ;HL=LAST LINE, SAME COLUMN
033D 19                 DAD     D
033E D1                 POP     D
```

```
033F 7E          FPRI:   MOV     A,M         ;GET CHAR.
0340 1F                  RAR                 ;UPPER NIBBLE TO LOWER
0341 1F                  RAR
0342 1F                  RAR
0343 1F                  RAR
0344 2F                  CMA                 ;COMPLEMENT A
0345 E6 0F               ANI     0FH         ;MASK OFF NIBBLE
0347 CD E6 03            CALL    PRINT       ;PRINT IT
034A D5                  PUSH    D
034B 11 C0 FF            LXI     D,-40H      ;HL=HL-1 LINE
034E 19                  DAD     D
034F D1                  POP     D
0350 3A 7A 04            LDA     COUNT       ;CHAR. COUNT=CHAR. COUNT-1
0353 3D                  DCR     A
0354 32 7A 04            STA     COUNT
0357 C2 3F 03            JNZ     FPRI        ;LOOP MORE IF NOT END OF LINE
035A E1                  POP     H           ;HL=ADDRESS AT TOP OF PIX
035B CD BA 03            CALL    SWAIT       ;WAIT FOR NEXT SECTOR
035E CD AB 03            CALL    CWAIT
0361 3E 03               MVI     A,STOP      ;STOP CARRAGE
0363 CD A4 03            CALL    COMND1
0366 CD C9 03            CALL    LF          ;LINEFEED
0369 3E 02               MVI     A,LEFT      ;START CARRAGE LEFT
036B CD A4 03            CALL    COMND1
036E 16 FF               MVI     D,0FFH      ;DIRECTION FOR PTINY ROUTINE(LEFT)
0370 CD BA 03            CALL    SWAIT       ;WAIT FOR FIRST SECTOR
0373 C3 78 03            JMP     $+5
0376 06 1E               MVI     B,1EH       ;CHANGE THIS BYTE TO LINE UP
                                             ;   THE LINES IN EACH DIRECTION
0378 CD AB 03  LOOP:     CALL    CWAIT       ;NOT IN THE SAME DIRECTION.
037B 05                  DCR     B
037C C2 78 03            JNZ     LOOP
037F 3E 80               MVI     A,80H       ;COUNT OF NUMBER OF CHARS. TO PRINT
0381 32 7A 04            STA     COUNT
0384 E5                  PUSH    H
0385 7E          RPRI:   MOV     A,M         ;GET CHAR
0386 2F                  CMA                 ;COMPLEMENT A
0387 E6 0F               ANI     0FH         ;MASK OFF NIBBLE
0389 CD E6 03            CALL    PRINT       ;PRINT IT
038C D5                  PUSH    D
038D 11 40 00            LXI     D,40H       ;ADD 1 LINE TO HL
0390 19                  DAD     D
0391 D1                  POP     D
0392 3A 7A 04            LDA     COUNT       ;CHAR COUNT= CHAR COUNT -1
0395 3D                  DCR     A
0396 32 7A 04            STA     COUNT
0399 C2 85 03            JNZ     RPRI        ;LOOP IF NOT END OF LINE
039C E1                  POP     H           ;HL=POINTER AT TOP OF SCREEN
039D 23                  INX     H
039E CD C9 03            CALL    LF          ;DO A LINE FEED
03A1 C3 08 03            JMP     HOME        ;GO HOME
03A4 D3 09       COMND1: OUT     CONTL       ;OUTPUT DATA TO COMMAND PORT
03A6 F6 80               ORI     80H
03A8 D3 09               OUT     CONTL
03AA C9                  RET
03AB DB 09       CWAIT:  IN      SENSE       ;WAIT FOR COLUMN HIGH
03AD E6 01               ANI     01H
03AF CA AB 03            JZ      CWAIT
```

```
0416 DB 09              IN      SENSE       ;WAIT FOR COLUMN HIGH
0418 E6 01              ANI     01H
041A CA 11 04           JZ      ONECL
041D 3E 0A              MVI     A,FIRE      ;FIRE PRINT HEAD
041F CD A4 03           CALL    COMND1
0422 DB 09      ENDCL:  IN      SENSE       ;WAIT FOR COLUMN LOW
0424 E6 01              ANI     01H
0426 C2 22 04           JNZ     ENDCL
0429 C9                 RET
                *
                *THIS IS THE TABLE USED TO PRINT THE GRAY SCALE.
                *IT IS ARRANGED 5 WIDE AND 6 BITS HIGH FROM D0
                *
042A 00 00 00   TABLE:  DB      00H,00H,00H,00H,00H     ;1
042D 00 00
042F 20 00 00           DB      20H,00H,00H,00H,00H     ;2
0432 00 00
0434 20 10 00           DB      20H,10H,00H,08H,00H     ;3
0437 08 00
0439 20 10 20           DB      20H,10H,20H,08H,04H     ;4
043C 08 04
043E 20 11 24           DB      20H,11H,24H,08H,04H     ;5
0441 08 04
0443 22 11 24           DB      22H,11H,24H,0AH,04H     ;6
0446 0A 04
0448 22 15 24           DB      22H,15H,24H,0BH,04H     ;7
044B 0B 04
044D 22 1D 24           DB      22H,1DH,24H,1BH,04H     ;8
0450 1B 04
0452 22 1D 24           DB      22H,1DH,24H,1BH,26H     ;9
0455 1B 26
0457 26 1D 26           DB      26H,1DH,26H,1BH,26H     ;10
045A 1B 26
045C 36 1D 36           DB      36H,1DH,36H,1BH,26H     ;11
045F 1B 26
0461 36 1D 36           DB      36H,1DH,36H,1BH,2FH     ;12
0464 1B 2F
0466 36 1F 36           DB      36H,1FH,36H,3BH,2FH     ;13
0469 3B 2F
046B 3E 1F 3F           DB      3EH,1FH,3FH,3BH,2FH     ;14
046E 3B 2F
0470 3F 1F 3F           DB      3FH,1FH,3FH,3BH,3FH     ;15
0473 3B 3F
0475 3F 3F 3F           DB      3FH,3FH,3FH,3FH,3FH     ;16
0478 3F 3F
047A            COUNT:  DS      1
047B            LINES:  DS      1
                        END

      0 ERROR(S) DETECTED
        LAST ADDRESS 047B


SYMBOL TABLE:

CLOSE   03DE    COLMN   040F    COLOP   03FD    COMND1  03A4    CONTL   0009
COUNT   047A    CW      03B2    CWAIT   03AB    DATA    0008    DISP    8000
```

ENDCL	0422	FIRE	000A	FPRI	033F	FRWD	03FB	FRWD2	0409
HOME	0308	HOME1	030D	LEFT	0002	LF	03C9	LFDLY	03D1
LFDLY1	03D5	LFLOP	03CB	LINES	047B	LOOP	0378	MPRINT	0300
ONECL	0411	PRINT	03E6	RIGHT	0001	RPRI	0385	SENSE	0009
STOP	0003	SW	03C1	SWAIT	03BA	TABLE	042A	TOF	0007
UP	0006	WARN	0103						

THE MICRO WORKS

DS-80 ENGINEERING CHANGE

Dear Customer,

We have discovered a problem in the DS-80 circuit design which causes errors in the sample and hold circuit. The SD-210 FET used as the sampling switch has its substrate floating. This was designed to reduce leakage current, but unfortunately it backfired. On some SD-210s the substrate will charge up to a few volts positive with respect to the source causing the FET to turn on; the gate loses control.

To correct this, simply connect a jumper wire from the substrate terminal to ground. This is *very* simple as there is a ground plane immediately adjacent to the uncommitted substrate pin on the SD-210.



P.O. BOX 1110 DEL MAR, CA. 92014 714-756-2687

Computer Pictures

1. THE VECTOR GRAPHICS, GRAPHICS BOARD MUST BE PLUGGED into the top memory BOARD of the computer, this BOARD MUST be a Vector Graphics 8K Memory BOARD (IT is plugged in through the 5 RIBBON connectors on the back of the BOARD)

2. Plug the 2 wire cable (COAXAL cable) into the MICROWORKS BOARD, Plug the CAMERA END into the camera, If the camera does NOT have a monitor plug the other wire into the monitor, This allows you to see exactly what the camera sees

3 Plug the vector GRAPHICS BOARD wire into the Vector Graphics BOARD AND the other END INTO * Monitor.

3.5 Change the LPT's cable-A TO the SIO PRINTER Port FOR the REC. - Plug the REC INTO the LPT ~~port~~ connection in to the back of the computer

4. LOAD the picture DISK SYSTEM into the computer. IN A 56K system this system MUST BE 32K (NOTE: only use the @100H switch when starting)

The Vector MEMORY BOARD MUST Be AT C000H AND FREE memory FROM B000H - BFFFH.

5 TYPE
A>DDT NSCAN.HEX (CR)
*(CR) ↑C
A>MBASIC Part.BAS (CR)
Now You're in MBASIC with the DRIVER PROG. Running

Memory

0000 —— AFFF – USER MEMORY

B000 – BFFF – WSCAN PROG

C000 – DFFF – Display Spirit 8K